
Fabriquer un Power router 

avec Philippe de Craene  dcphilippe@yahoo.fr 

et Dominique Boucherie  ptiwatt@mailoo.org 

Date : Juillet 2018 

 

Un power router est un dispositif qui absorbe le trop d’énergie produit localement – soit par une éolienne, ou par 

panneaux photovoltaïques, ou tout autre équipement – afin d’éviter d’injecter cette énergie sur le réseau public. 

Ce document décrit comment réaliser cet appareil. 

 

A noter : la réalisation de ce programme a demandé plusieurs dizaines d’heures de développement, apprendre à 

utiliser l’Arduino, comprendre son comportement, en cramer deux… apprendre son langage, faire des tests sur 

différents capteurs de courant, tester différents algorithmes, et imaginer tous les tests possibles pour fiabiliser au 

maximum l’appareil. 

Toute contribution en vue de l’amélioration de l’appareil est la bienvenue ! Il vous est juste demandé de conserver 

mon nom et mon email dans l’entête du programme, et bien sûr de partager avec moi cette amélioration. Merci. 

  

mailto:dcphilippe@yahoo.fr
mailto:ptiwatt@mailoo.org

Introduction 
Il s’agit de mesurer la tension et le courant 

après le compteur d’arrivée de l’énergie 

public : en mode de fonctionnement 

classique, nous avons plus ou moins 

(rappelez-vous des cours d’électricité au 

lycée et ce fameux cos ϕ) la tension en 

phase avec le courant. Or lorsqu’il y a 

injection, la tension et le courant seront en 

inversion de phase. 

Il s’agit donc de détecter si le rapport 

courant/tension sont oui ou non en 

opposition de phase, auquel cas il s’agira 

d’alimenter un appareil composé d’une 

résistance pure telle qu’un chauffe-eau ou 

un halogène en proportion du courant 

produit afin de ne plus injecter dans le 

réseau public. 

L’appareil est composé d’un élément de 

mesure du courant (mini pince 

ampèremétrique), d’un élément de mesure de tension (simple transformateur bobiné), d’un élément de commande 

par triac, le tout piloté par un programme dans un Arduino Uno R3. 

La partie hardware est composée de 3 parties :  

- La carte Arduino avec son câble USB et/ou son alimentation 5V  

- La carte d’extension avec les 2 capteurs 

- La carte du module triac 

La partie logicielle est composée de 3 parties : 

- La gestion des données tensions et courants pour détecter l’injection ou la consommation 

- La gestion du triac 

- La gestion de l’asservissement qui dose la conductivité dans la résistance afin d’empêcher l’injection sur le 

réseau public 

L’appareil proposé ici convient pour quelques centaines de Watts à absorber. L’élément limitateur est le Triac qui 

dans notre cas accepte 5A maxi. 

Liste des courses 
 

1- Un arduino Uno R3 : https://fr.aliexpress.com/item/One-set-New-2016-UNO-R3-ATmega328P-CH340G-

MicroUSB-Compatible-for-Arduino-UNO-Rev-3-0/32696412561.html?spm=a2g0s.9042311.0.0.27426c37rrsgNO 

 

2- Une carte d’extension : https://fr.aliexpress.com/item/Free-Shipping-UNO-Proto-Shield-prototype-

expansion-board-with-SYB-170-mini-bread-board-based-

For/32502867722.html?spm=a2g0s.9042311.0.0.27426c37rrsgNO 

 

3- Un capteur de courant sensible : https://fr.aliexpress.com/item/Free-shipping-0-30A-sensor-split-core-

current-transformer-SCT006/32579590465.html?spm=a2g0s.9042311.0.0.27426c37zd6RJS 

 

https://fr.aliexpress.com/item/One-set-New-2016-UNO-R3-ATmega328P-CH340G-MicroUSB-Compatible-for-Arduino-UNO-Rev-3-0/32696412561.html?spm=a2g0s.9042311.0.0.27426c37rrsgNO
https://fr.aliexpress.com/item/One-set-New-2016-UNO-R3-ATmega328P-CH340G-MicroUSB-Compatible-for-Arduino-UNO-Rev-3-0/32696412561.html?spm=a2g0s.9042311.0.0.27426c37rrsgNO
https://fr.aliexpress.com/item/Free-Shipping-UNO-Proto-Shield-prototype-expansion-board-with-SYB-170-mini-bread-board-based-For/32502867722.html?spm=a2g0s.9042311.0.0.27426c37rrsgNO
https://fr.aliexpress.com/item/Free-Shipping-UNO-Proto-Shield-prototype-expansion-board-with-SYB-170-mini-bread-board-based-For/32502867722.html?spm=a2g0s.9042311.0.0.27426c37rrsgNO
https://fr.aliexpress.com/item/Free-Shipping-UNO-Proto-Shield-prototype-expansion-board-with-SYB-170-mini-bread-board-based-For/32502867722.html?spm=a2g0s.9042311.0.0.27426c37rrsgNO
https://fr.aliexpress.com/item/Free-shipping-0-30A-sensor-split-core-current-transformer-SCT006/32579590465.html?spm=a2g0s.9042311.0.0.27426c37zd6RJS
https://fr.aliexpress.com/item/Free-shipping-0-30A-sensor-split-core-current-transformer-SCT006/32579590465.html?spm=a2g0s.9042311.0.0.27426c37zd6RJS

Le capteur de courant est une sorte de mini transformateur bobiné qui doit 

être connecté au secondaire sur une résistance (dite de Burden) de 100Ω. 

Inutile d’augmenter la valeur de cette résistance pour tenter d’obtenir un 

meilleur taux de transformation, ça sature très vite ces petits capteurs. 

Au niveau de votre installation domestique, si vous êtes moderne durable et 

bio vous ne consommez pas plus de 4000W en instantané. Ce qui donne un 

courant I = 17A maxi.  

Donc laissez tomber les gros capteurs de 30A qui sont perdus pour mesurer 

les faibles courants. Plus vous trouverez un capteur au ratio 

Itraversé/Isecondaire élevé, plus votre mesure sera sensible, et meilleurs sera 

votre PowerRouter. 

 

4- Un capteur de tension : un transformateur bobiné (il faut absolument 

une tension alternative) délivrant de l’ordre de 4V crête à crête au secondaire (généralement ce qui est disponible 

est bien plus, on calculera alors un pont diviseur de tension (cf plus bas). 

 

5- Un module de commande de Triac : https://fr.aliexpress.com/item/AC-Light-Dimmer-Module-for-PWM-

control-1-Channel-3-3V-5V-logic-AC-50-60hz/32802025086.html?spm=a2g0s.9042311.0.0.27426c37xh5Y5t 

Ce module est très pratique : il comporte ET le détecteur de passage à 

zéro nécessaire à l’utilisation du triac, et le dispositif de commande du 

triac, tout cela par des optocoupleurs qui assurent l’isolation de la haute 

tension du secteur. 

Ce modèle est un peu limité en puissance, 5A maxi, 2A nominal… 

Cependant le triac est donné pour 16A. La limitation doit être dû à la 

faible épaisseur des pistes du circuit imprimé : il suffit dans ce cas de 

doubler les pistes de puissance par du fil électrique  

 

6- Une alimentation 5V pour l’Arduino, c’est-à-dire un chargeur de 

téléphone portable de récupération, du câble Dupont (bof bof c’est plein 

de mauvais contacts, rien de vaut mieux que la soudure), une paire de 

prises Cinch (mâle + femelle châssis) pour connecter les capteurs, et une toute petite poignée de composants 

électriques dont la liste est un peu plus bas. 

 

7- Une jolie boite pour intégrer durablement le 

montage (c’est d’ailleurs l’élément le très loin le plus cher) 

 

 

 

 

 

  

https://fr.aliexpress.com/item/AC-Light-Dimmer-Module-for-PWM-control-1-Channel-3-3V-5V-logic-AC-50-60hz/32802025086.html?spm=a2g0s.9042311.0.0.27426c37xh5Y5t
https://fr.aliexpress.com/item/AC-Light-Dimmer-Module-for-PWM-control-1-Channel-3-3V-5V-logic-AC-50-60hz/32802025086.html?spm=a2g0s.9042311.0.0.27426c37xh5Y5t

Circuit électrique autour des capteurs 
 

Les 2 capteurs fournissent chacun un signal sinusoïdal alternatif, qu’il faut adapter pour l’Arduino. En effet tout signal 

entrant sur l’Arduino doit rigoureusement être compris entre 0 et 5V. D’où l’idée de connecter les capteurs à mi-

tension d’alimentation, soit 2,5V, en vérifiant que l’amplitude des signaux que délivreront les capteurs ne dépassent 

jamais le +5V ou deviennent négatives. 

Mesure de tension 
 

 

Il faut calculer R2 afin d’avoir R2 = Vtransfo eff *0,7 -1. Mesurer la tension à vide du transformateur avec un 
multimètre fournit la tension efficace. 

R3 = R4, peu importe la valeur entre 1 et 47k 
C1 entre 1 et 47uF. 
2 zeners de 4,7V montées tête-bêche assurent une protection en cas de surtension. 

La sortie de ce montage (Arduino input sur le schéma) est à relié à l’entrée A1 de l’Arduino . 

 

Mesure de courant 

R1 = R2, peu importe la valeur entre 1 et 47k 
C1 entre 1 et 47uF. 
2 zeners de 4,7V montées tête-bêche assurent une protection en cas de surtension. 

La sortie de ce montage (Arduino input sur le schéma) est à relié à l’entrée A0 de l’Arduino . 

Quelques photos 
 

On notera la présence de picots en A B C D E (en jaune sur la photo) pour le raccordement à l’aide de câbles Dupont 

Premier test 
 

1- Sans brancher les capteurs, enficher la carte d’extension dans l’Arduino, brancher l’Arduino sur le port USB 

du PC : Mesurer 2,5V sur les 2 bornes A et B (cf photo ci-dessus) 

2- Brancher le capteur de courant entre A et C, et le capteur de tension entre B et E. 

3- Connecter C en entrée analogique 0, et D en entrée analogique 1. 

 

4- On connecte l’arduino au PC avec un câble USB. 

5- Installer l’interface de programmation, le programme est téléchargeable ici : 

https://www.arduino.cc/en/Main/OldSoftwareReleases 

6- On installe les drivers pour l’arduino UNO : http://283.mytrademe.info/ch340.html 

7- On lance le programme Arduino : 

1- Menu outils-> type de carte-> UNO 

2- Menu outils-> PORT-> ComX ou X représente le port sur lequel est installé votre arduino. 

3- On efface ce qu’il y a dans la fenêtre d’édition 

4- On y colle le code ci-dessous : 

 
// minMaxAndRangeChecker 
// A simple tool to investigate the ADC values that are seen at the 
// first four analogue inputs of an Atmega chip, as used on an emonTx 
// 
// Robin Emley (calypso_rae on the Open Energy Monitor forum) 
// 
// 20th April 2013 
// 
 
int val_a0, val_a1, val_a2, val_a3; 
int minVal_a0, minVal_a1, minVal_a2, minVal_a3; 
int maxVal_a0, maxVal_a1, maxVal_a2, maxVal_a3; 
 
int loopCount = 0; 
unsigned long timeAtLastDisplay = 0; 
byte displayLineCounter = 0; 

https://www.arduino.cc/en/Main/OldSoftwareReleases
http://283.mytrademe.info/ch340.html

void setup(void) { 
  Serial.begin(9600); 
  Serial.print("ready ..."); 
  delay(700); 
  Serial.println (); 
  Serial.println(" The Min, Max and Range ADC values for analog inputs 0 to 3:"); 
} 
 
void loop(void) { 
  val_a0 = analogRead(0); //  CT2 
  val_a1 = analogRead(1); //  CT3 
  val_a2 = analogRead(2); //  Vsensor 
  val_a3 = analogRead(3); //  CT1 
 
  if (val_a0 < minVal_a0) { minVal_a0 = val_a0;} 
  if (val_a0 > maxVal_a0) { maxVal_a0 = val_a0;} 
  if (val_a1 < minVal_a1) { minVal_a1 = val_a1;} 
  if (val_a1 > maxVal_a1) { maxVal_a1 = val_a1;} 
  if (val_a2 < minVal_a2) { minVal_a2 = val_a2;} 
  if (val_a2 > maxVal_a2) { maxVal_a2 = val_a2;} 
  if (val_a3 < minVal_a3) { minVal_a3 = val_a3;} 
  if (val_a3 > maxVal_a3) { maxVal_a3 = val_a3;} 
 
  unsigned long timeNow = millis(); 
  if ((timeNow - timeAtLastDisplay) >= 3000) { 
    timeAtLastDisplay = timeNow; 
    displayVal(minVal_a0); 
    displayVal(maxVal_a0); 
    displayVal(maxVal_a0 - minVal_a0); 
    Serial.print(";  "); 
 
    displayVal(minVal_a1); 
    displayVal(maxVal_a1); 
    displayVal(maxVal_a1 - minVal_a1); 
    Serial.print(";  "); 
 
    displayVal(minVal_a2); 
    displayVal(maxVal_a2); 
    displayVal(maxVal_a2 - minVal_a2); 
    Serial.print(";  "); 
 
    displayVal(minVal_a3); 
    displayVal(maxVal_a3); 
    displayVal(maxVal_a3 - minVal_a3); 
    Serial.println(); 
 
    resetMinAndMaxValues();  
    displayLineCounter++; 
 
    if (displayLineCounter >= 5) { 
      Serial.println(); 
      displayLineCounter = 0; 
      delay(2000); // to allow time for data to be accessed 
    } 
  }  
} 
 
void resetMinAndMaxValues() { 
  minVal_a0 = 1023, minVal_a1 = 1023, minVal_a2 = 1023, minVal_a3 = 1023; 
  maxVal_a0 = 0, maxVal_a1 = 0, maxVal_a2 = 0, maxVal_a3 = 0; 
} 
void displayVal(int intVal){ 
  char strVal[4]; 
  byte lenOfStrVal; 
  itoa(intVal, strVal, 10); // decimal conversion to string 
  lenOfStrVal = strlen(strVal); // determine length of string 
 
  for (int i = 0; i < (4 - lenOfStrVal); i++)  { 
    Serial.print(' '); 
  } 
 
  Serial.print(strVal); 
} 

 

Enregistrer ce programme sur le PC sous le nom de testminmax.ino par exemple. 
Puis : Menu croquis -> téléverser. 
Ouvrir le moniteur série :  Menu outils -> moniteur série 
Une autre fenêtre s’ouvre. Régler le débit (baud rate à 9600 si ce n’est pas le réglage par défaut) 

Dans un premier temps on ne bascule pas l’inter 230V de la multiprise. L’arduino est uniquement alimenté (en 5V) 
par le port USB du PC et les valeurs de l’entrée tension et l’entrée courant sont donc égale à 0. 
On voit apparaitre des valeurs dans le moniteur série : 

Entrée A0 

Première colonne = min en bits (de 0 à 1023) 

Deuxième colonne = max (de 0 à 1023) 

Troisième colonne = écart (max-min) 

Entrée A1 

Première colonne = min en bits (de 0 à 1023) 

Deuxième colonne = max (de 0 à 1023) 

Troisième colonne = écart (max-min) 

Entrée A2 Etc etc 

On doit lire des valeurs proches de 512 pour A0 at A1. 

510 514 4   508 511 3 

Les entrées analogiques passent par un convertisseur numérique qui fournit 0 à 1024 bits pour une valeur lue de 0 à 
5V. A repos nous obtenons donc 512 bits, aux tolérances des valeurs de résistances près (5%). 
Sinon c’est qu’il y a un problème sur le pont diviseur de tension de la voie concernée  
A remarquer que le bruit des convertisseurs est de l’ordre de 4 bits. 

Ensuite tester un appareil genre lampe halogène d’une centaine de Watts. Les valeurs doivent changer et se 
comporter à peu près comme ceci : 

Sur A0 : 415 606 191 

 Sur A1 : 110 912 802 

C’est tout bon ? parfait. 

Le module triac 
 

Le but du « jeu » est d’ouvrir plus ou moins un interrupteur pour déverser sur une résistance le trop d’énergie 

produit localement, afin de ne pas injecter sur le réseau public. 

Principe de fonctionnement 
 

L’usage d’un triac est 

idéal car il s’éteint 

tout seul à chaque 

passage à zéro de la 

sinusoïde de tension. 

Entre chaque passage à 

zéro il peut être 

déclenché (le fire) et 

plus c’est tôt plus il 

sera conducteur pour la 

résistance, et 

inversement. Nous 

pouvons donc 

entrevoir qu’à chaque 

demi-alternance le 

programme va 

déclencher au bon 

moment pour juste 

délester le trop 

d’énergie produit, et 

surtout ne rien déverser lorsque ce n’est pas nécessaire. 

Par rapport au schéma, à chaque demi-alternance t3, à un instant t1 variable on procède au fire t2 qui rend le triac 

conducteur, jusqu’au prochain passage à zéro de la tension. 

Ce dispositif oblige 2 contraintes : 

- Il faut un procédé pour détecter chaque passage à zéro. Pour l’instant sur l’Arduino nous n’avons comme 

mesure de tension qu’un nombre de bit de 0 à 1024. Il serait possible de créer un programme pour détecter chaque 

passage à zéro, mais ce genre de boucle prendrait beaucoup de ressources, ressources qui vont servir à bien autre 

chose…. 

Il va donc falloir un circuit supplémentaire pour détecter les passages à zéro. 

- Le triac fonctionne sur le secteur, soit 230V, il est nécessaire de l’isoler pour que la phase ne se « balade » 

pas sur l’Arduino. 

Le module triac va répondre à ces 2 besoins et reprend le schéma ci-dessous, avec l’usage d’optocoupleurs : 

Il existe 2 familles d’optocoupleurs « simples » pour détecter le passage à zéro  : les modèles avec 2 LEDs tête-bêche 

qui vont bien détecter chaque demi-alternance, et les modèles simple LED, auquel cas il faudra simplement ajouter 

en amont de la LED un pont redresseur de 4 diodes. 

 

La photo du montage 
 

 

 

On remarquera : 

- A : la connexion pour la sonde de courant 

- B : la connexion pour la sonde de tension 

- C : les autres entrées analogiques sont reliées à la masse 

- D : le fil du zéro-détecte sur l’entrée 2  

- E : le fil de commande du triac sur la sortie 10 

- F : bornier vert = arrivée 230V, bornier bleu = charge résistive 

Deuxième test 
 

L’astuce du programme est de gérer une interruption : le programme fait sa cuisine… et toutes les 10ms (pour un 

secteur à 50Hz) il reçoit une info sur l’entrée 2. Il arrête donc de touiller sa cuisine pour gérer cette interruption. 

La 2ème astuce du programme est de définir un compteur système de 78us : 128 fois par 10ms (et donc par demi-

cycle secteur) il sait qu’il va falloir s’occuper de quelque chose, à savoir d’un compteur temporel qui définie le 

moment du fire du triac. 

Ces 2 astuces font appel à une librairie qu’il faut installer sur le PC : TimerOne.h 

A voir ici : http://playground.arduino.cc/Code/Timer 

1- Télécharger la librairie https://github.com/JChristensen/Timer/archive/v2.1.zip 

2- La déclarer : (In the Arduino IDE) Sketch > Include Library > Add .ZIP Library > select the downloaded file > 

Open 

3- Créer un nouveau programme avec le code ci-dessous : 

/* 
AC Light Control 
 
  Philippe de Craene <dcphilippe@yahoo.fr> / May 2018 
  merci à Ryan McLaughlin <ryanjmclaughlin@gmail.com>  
  pour la partie commande du triac 
  source : https://web.archive.org/web/20091121155320/http://www.arduino.cc:80/cgi-
bin/yabb2/YaBB.pl?num=1230333861/0 
 
*/ 
 
#include <TimerOne.h>           // Avaiable from ttp://www.arduino.cc/playground/Code/Timer1 
 
volatile int i=0;               // Variable to use as a counter 
volatile boolean zero_cross=0;  // Boolean to store a "switch" to tell us if we have crossed 
zero 
int triac_pin = 10;             // Output to Opto Triac 
int Dimmer_pin = 0;             // Pot for testing the dimming 
int LED = 3;                    // LED for testing 
int dim = 0, dimmax = 128;      // Dimming level (0-128)  0 = on, 128 = 0ff 
int freqStep = 75;   // Set the delay for the frequency of power (65 for 60Hz, 78 for 50Hz) per 
step (using 128 steps) 
                     // freqStep may need some adjustment depending on your power the formula 
                     // you need to us is (500000/AC_freq)/NumSteps = freqStep 
                     // You could also write a seperate function to determine the freq 
 
void setup() {                                     // Begin setup 
 pinMode(triac_pin, OUTPUT);                       // Set the Triac pin as output 
 pinMode(LED, OUTPUT);                             // Set the LED pin as output 
 attachInterrupt(0, zero_cross_detect, FALLING);   // Attach an Interupt to Pin 2 (interupt 0) 
for Zero Cross Detection 
 Timer1.initialize(freqStep);                      // Initialize TimerOne library for the freq 
we need 
 Timer1.attachInterrupt(dim_check, freqStep);      // Use the TimerOne Library to attach an 
interrupt 
                                                   // to the function we use to check to see if 
it is 
                                                   // the right time to fire the triac.  This 
function 
                                                   // will now run every freqStep in 
microseconds.                                         
 Serial.begin(9600); 
 Serial.print("ready ..."); 
 delay(300); 
 Serial.println ();     
}                                                  // End setup 
 
void zero_cross_detect() {          // function to be fired at the zero crossing 
   zero_cross = 1;                  // set the boolean to true to tell our dimming function that 
a zero cross has occured 
}                                   // End zero_cross_detect 
 
void dim_check() {                  // Function will fire the triac at the proper time 
 if(zero_cross == 1) {   // First check to make sure the zero-cross has happened else do nothing 
   if(i>=dim) {                     // Check and see if i has accumilated to the dimming value 
we want 
     digitalWrite(triac_pin, HIGH);    // Fire the Triac mid-phase 
     delayMicroseconds(50);            // Pause briefly to ensure the triac turned on 
     digitalWrite(triac_pin, LOW);     // Turn off the Triac gate  
                                       //(Triac will not turn off until next zero cross) 
     i = 0;                         // Reset the accumilator 

http://playground.arduino.cc/Code/Timer
https://github.com/JChristensen/Timer/archive/v2.1.zip

     zero_cross = 0;                // Reset the zero_cross so it may be turned on again at the 
next zero_cross_detect     
   } else { 
     i++;                    // If the dimming value has not been reached, incriment our counter 
   }                         // End dim check 
 }                           // End zero_cross check 
}                            // End dim_check function 
 
void loop() {                         // Main Loop 
 while (dim> -1 ) { 
       for (dim =0 ; dim <= dimmax ; dim++) { 
           Serial.print(" valeur de dim = "); 
           Serial.println(dim); 
           analogWrite(LED, dim);             // Write the value to the LED for testing 
           delay(100); 
       } 
       for (dim = dimmax ; dim > 0 ; dim--) { 
           Serial.print(" valeur de dim = "); 
           Serial.println(dim); 
           analogWrite(LED, dim);             // Write the value to the LED for testing 
           delay(100); 
       } 
 } 
} 

 

Enregistrer ce programme sur le PC sous le nom de testtriac.ino par exemple. 
Puis : Menu croquis -> téléverser. 
Ouvrir le moniteur série :  Menu outils -> moniteur série 
Une autre fenêtre s’ouvre. Il doit s’afficher la valeur de dim qui s’incrémente puis se décrémente de 0 à 128. 

En branchant une ampoule à filament ou tout autre éclairage dimable celui-ci doit s’allumer puis s’éteindre 

progressivement, et infiniment. 

Ça marche ? on continue. 

Mesure et traitement de la tension et du courant 
 

Nous savons que la différence entre le mode injection et le mode consommation est la phase entre tension et 

courant, sachant que celle-ci se trouve en opposition de phase en mode injection. 

 

mode consommation    mode injection 

 

Le comment faire  
 

Il serait possible de mesure cette phase avec un : 
valeurV = analogRead(voltageSensorPin) 

suivi d’un : 
valeurI = analogRead(currentSensorPin) 

Cependant il faut attendre un moment opportun pour réaliser cette opération, par exemple en effectuant la lecture 

du courant au moment où la tension atteint un seuil. Le problème est qu’il faut lire sans arrêt (en fait le plus vite 

possible) mesurer la tension dans l’espoir tomber sur le seuil qui autorisera la mesure de I. Tout cela oblige au 

recours d’une boucle de test qui prend du temps, temps qui doit servir à autre chose. 

Ensuite il se pose 2 autres problèmes, les jolies sinusoïdes ci-dessus sont du monde des bisounours, au moins pour 

celle du courant : 

- D’abord il y a toujours un déphasage entre courant et tension, ce fameux cos ϕ, parce que nos équipements 

ne sont que rarement des résistances pures, il y a des moteurs (selfique) des Leds ou du fluorescent (capacitif) qui 

font que le courant sera toujours en avance ou en retard de phase par rapport à la tension. Le résultat de la mesure 

ponctuelle donne donc une valeur de courant qui ne correspond pas à la réalité du courant consommé. 

- Pire ! Il y a quantité d’appareils qui ne fonctionnent pas en régime linéaire, mais en régime impulsionnel : Il y 

a des alimentations à découpage (télévision, ordinateur) qui « s’amusent » à découper la tension – en donc le 

courant – en tronçons à des vitesses aussi rapides que variables, et les triacs (lave-linge, le Power Router) qui, lors du 

fire fait un appel en courant qui fait comme des cornes sur la courbe de courant, et ce à chaque demi-période 

secteur…. 

La mesure ponctuelle n’étant pas du tout adaptée, nous allons utiliser la même méthode que les programmes qui 

transforment l’Arduino en appareil de mesure de tensions alternatives : un procédé qui fait une acquisition sur un 

laps de temps (paramétrable) et produit un résultat cohérent après traitement. 

La solution utilisée et le troisième test 
 

La mesure du courant alternatif – et aussi de tous les autres paramètres : tension alternative, puissance utile, 

puissante apparente, facteur de puissance - fait appel à une librairie qu’il faut installer sur le PC : EmonLib.h 

A voir ici : https://learn.openenergymonitor.org/electricity-monitoring/ctac/how-to-build-an-arduino-energy-

monitor 

1- Télécharger la librairie https://github.com/openenergymonitor/EmonLib/archive/master.zip 

2- La déclarer : (In the Arduino IDE) Sketch > Include Library > Add .ZIP Library > select the downloaded file > 

Open 

3- Créer un nouveau programme avec le code ci-dessous : 

 

// EmonLibrary examples openenergymonitor.org, Licence GNU GPL V3 
 
#include "EmonLib.h"             // Include Emon Library 
EnergyMonitor emon1;             // Create an instance 
 
void setup() 
{   
  Serial.begin(9600); 
   
  emon1.voltage(1, 155, 1.7);  // Voltage: input pin, calibration, phase_shift 
  emon1.current(0, 105);       // Current: input pin, calibration. 
 
  
 Serial.println("ready ..."); 
 Serial.println (); 
 delay(500);  
 Serial.println("   V   /   I   /  Peff   /   Pi    (en Volts/Ampères/Watts/VA/)"); 
 Serial.println(); 
} 
 
void loop() 
{ 
  emon1.calcVI(20,200);         // Calculate all. No.of half wavelengths (crossings), time-out 
//  emon1.serialprint();           // Print out all variables (realpower, apparent power, Vrms, 
Irms, power factor) 
   
  float realPower       = emon1.realPower;        //extract Real Power into variable 
  float apparentPower   = emon1.apparentPower;    //extract Apparent Power into variable 
  float powerFactor     = emon1.powerFactor;      //extract Power Factor into Variable 
  float supplyVoltage   = emon1.Vrms;             //extract Vrms into Variable 
  float Irms            = emon1.Irms;             //extract Irms into Variable 
 
    Serial.print("   "); 
    Serial.print(supplyVoltage); 
    Serial.print("  |  "); 
    Serial.print(Irms); 
    Serial.print("  |  "); 
    Serial.print(realPower); 
    Serial.print("  |  "); 
    Serial.print(apparentPower); 
    Serial.print("  |  "); 
    Serial.println(powerFactor); 
} 

 

Enregistrer ce programme sur le PC sous le nom de testmesuresAC.ino par exemple. 
Puis : Menu croquis -> téléverser. 
Ouvrir le moniteur série :  Menu outils -> moniteur série 
Une autre fenêtre s’ouvre. Il doit s’afficher la valeur de la tension, du courant, de la puissance utile, de la puissance 
apparente, du facteur de puissance. 

En effectuant la mesure sur un appareil en marche, il est possible de régler les 2 paramètres de calibration pour : 

emon1.voltage(1, 155, 1.7) 

ici c’est 155, choisir une valeur pour s’approcher d’une lecture de 230V 

emon1.current(0, 105) 

ici c’est 105, idem choisir une valeur pour que la valeur de puissance utile s’approche de la puissance réelle 

consommée.  

https://learn.openenergymonitor.org/electricity-monitoring/ctac/how-to-build-an-arduino-energy-monitor
https://learn.openenergymonitor.org/electricity-monitoring/ctac/how-to-build-an-arduino-energy-monitor

Le programme final 
 

Il suffit de rassembler les morceaux dédiés au fonctionnement du triac et de la lecture du courant, puis d’ajouter le 

traitement du délai de déclenchement du triac en fonction de la puissance injectée. 

 

/* 
 
Power_Router est un système qui permet d'utiliser l'excédent d'énergie autoproduit par  
l'allumage d'un appareil résistif (facteur de puissance proche de 1) ce qui évite l'injection  
au réseau public de distribution d'électricité. 
 
Le principe de fonctionnement est le suivant : 
- détection de phase entre courant et tension permet de savoir si on consomme ou bien on injecte 
- en cas d'injection il se produit la mise en route progressive d'un dispositif d'absorption  
d'excédent de puissance  
- la mesure du courant permet d'ajuster au mieux le niveau d'absorption de cet excédent. 
 
le programme prévoit : 
- une sonde de tension : simple transfo 230V/5V Crête à Crête sur mi-tension (2.5V) 
- une sonde de courant : 20A/25mA sur mi-tension (2.5V) 
- un module de commande par triac 
- un dispositif de détection de passage à zéro de la sinusoïde de tension secteur (par exemple  
l'optocoupleur H11AA1) 
- la bibliothèque TimeOne.h à installer et disponible là :  
http://www.arduino.cc/playground/Code/Timer 
- la bibliothèque EmonLib.h à installer et disponible là :  
https://github.com/openenergymonitor/EmonLib/archive/master.zip 
 
La gamme de puissances testée est va de 300 à 1000W 
 
merci à Ryan McLaughlin <ryanjmclaughlin@gmail.com> pour avoir étudié et mis au point la partie  
commande du triac  
il y a quelques années et que j'ai repris dans ce programme :) 
source : https://web.archive.org/web/20091212193047/http://www.arduino.cc:80/cgi-
bin/yabb2/YaBB.pl?num=1230333861/15 
 
_________________________________________________________________ 
|                                                               | 
|       auteur : Philippe de Craene <dcphilippe@yahoo.fr        | 
|           pour l' Association P'TITWATT                       | 
_________________________________________________________________ 
 
Toute contribution en vue de l’amélioration de l’appareil est la bienvenue ! Il vous est juste 
demandé de conserver mon nom et mon email dans l’entête du programme, et bien sûr de partager  
avec moi cette amélioration. Merci. 
 
chronologie des versions : 
version 0.5 - 3 mai 2018     - boucle de décrémentation dim -- 
version 0.8 - 5 juil. 2018   - 1ère version fonctionnelle, pb du pic de courant du triac  
version 1   - 6 juil. 2018   - ajout de la bibliothèque EmonLib.h pour mesure du secteur 
version 1.4 - 7 juil. 2018   - simplification des tests sur sPower et dim. Version film youtube 
version 1.6 - 8 juil. 2018   - ajout LED d'overflow + optimisation des paramètres + seuilPoff 
 
*/ 
 
#include <TimerOne.h>           // librairie à installer depuis 
http://www.arduino.cc/playground/Code/Timer1 
#include <EmonLib.h>            // librairie à installer depuis 
https://github.com/openenergymonitor/EmonLib/archive/master.zip 
 
EnergyMonitor emon1;            // Create an instance 
 
// détermination des entrées / sorties : 
 
byte triac_pin         = 10;    // sortie numérique pour le Triac 
byte triacLED          = 3;     // sortie numérique pour la LED de test du triac 
byte limiteLED         = 13;    // sortie numérique pour la LED d'overflow 
byte voltageSensorPin  = 1;     // détecteur de tension entrée analogique 1 
byte currentSensorPin  = 0;     // détecteur de courant entrée analogique 0 
byte zeroCrossPin      = 2;     // détecteur de phase entrée digitale 2 
 
// variables de gestion des interruptions (zero-crossing) : 
  
int dimmax = 128;          // valeur max de dim pour inhiber le triac 
int dim = dimmax;          // Dimming level (0-128)  0 = on, 128 = 0ff  
char periodStep = 76;      // détermine la période du timer (65 pour 60Hz, 78 pour 50Hz, en µs) 
                           // suivant la formule (500000/AC_freq)/NumSteps = periodStep  

                           // 78*128=10ms=1/2 période 50Hz mais aux tests 76 marche mieux 
 
volatile int i = 0;             // Variable to use as a counter 
volatile boolean zero_cross=0;  // Boolean to store a "switch" to tell us if we have crossed 
zero 
 
// variables de gestion de la mesure de puissance (seuls valeurI et rPower nous intéressent) : 
 
long valeurI;                   // extract Irms into Variable (long supprime la « , ») 
long rPower;                    // extract Real Power into variable 
 
// variable de calibration 
                                // valeurs en mW (milliwatts) qui déterminent les seuils :  
int seuilP = 3000;              // l'hystérésis d'asservissement : 3000 = 3W => hystérésis à 6W 
int seuilPoff = 25000;          // seuil d'arrêt instantané en cas de chute de production 
 
 
// 
// SETUP 
//_____________________________________________________________________________________________ 
 
void setup() {                  // Begin setup 
 pinMode(triac_pin, OUTPUT);    // Set the Triac pin as output 
 pinMode(triacLED, OUTPUT);     // Set the LED pin as output 
 pinMode(limiteLED, OUTPUT);      
 
attachInterrupt(digitalPinToInterrupt(zeroCrossPin), zero_cross_detect, RISING);   
                            // en cas d'interruption lance 'zero_cross_detect' en mode RISING 
                            // à chaque changement de LOW à HIGH de zeroCrossPin.  
// documentation : https://www.arduino.cc/reference/en/language/functions/external-
interrupts/attachinterrupt/ 
 
 Timer1.initialize(periodStep);                    // initialisation de la librairie TimerOne 
 Timer1.attachInterrupt(dim_check, periodStep);    // Use the TimerOne Library to attach an 
interrupt 
                                           // to the function we use to check to see if it is 
                                           // the right time to fire the triac.  This function 
                                           // will now run every periodStep in microseconds. 
 
emon1.voltage(voltageSensorPin, 155, 1.7);      // Voltage: input pin, calibration, phase_shift 
emon1.current(currentSensorPin, 8000);          // Current: input pin, calibration  
 
// comparer l'affichage de la console avec celui d’un wattmètre pour s'approcher de la réalité.  
// Sinon faire le test avec une charge connue, par exemple une ampoule tungstène de 60W 
// Les valeurs ne sont pas critiques, 155 pour la tension donne environ 230V,  
// 8000 donne le courant en mA (milliampères) 
 
 Serial.begin(9600); 
 Serial.println (); 
 Serial.println("ready ..."); 
 Serial.println (); 
 delay(500);  
 Serial.print("   I (mA)  ||   Pu (mw)  |  Pa (mVA) ||"); 
// Serial.print("    FP    ||"); 
 Serial.println("   dim   |    i    )"); 
 Serial.println(); 
}                               // End setup 
 
 
// 
// ZERO CROSS DETECT : gestion du passage à zéro par interruption 
//____________________________________________________________________________________________ 
 
void zero_cross_detect() {       // function to be fired at the zero crossing  
   zero_cross = 1;               // set the boolean to true to tell our dimming function that a 
zero cross has occured 
}                                // End zero_cross_detect 
 
 
// 
// DIM CHECK : gestion du triac par interruption 
//____________________________________________________________________________________________ 
 
void dim_check() {                        // Function will fire the triac at the proper time 
 if(zero_cross == 1 && dim < dimmax) {    // First check to make sure the zero-cross has  
                                          // happened else do nothing 
                                          // ET inutile de compter le fire ça doit être éteint 
     if(i>=dim) {           // i est un compteur qui détermine le retard au fire. plus dim  
                            // est élevé, plus de temps prendra le compteur i et plus tard 
       digitalWrite(triac_pin, HIGH);     // se fera le fire du triac  
       delayMicroseconds(50);             // Pause briefly to ensure the triac turned on 
       digitalWrite(triac_pin, LOW);      // Turn off the Triac gate, le triac reste conducteur 
       i = 0;                             // Reset the accumilator 
       zero_cross = 0;                    // Reset the zero_cross so it may be turned on again 
at the next zero_cross_detect     
       }  
    else { i++; }           // If the dimming value has not been reached, incriment our counter 

 }                          // End zero_cross check 
}                           // End dim_check function 
 
 
// 
// LOOP : programme principal (qui tourne en boucle) 
//____________________________________________________________________________________________ 
 
void loop() {                  // Main Loop 
 
// Ci-dessous est le relevé des mesures du courant et de la puissance utile 
// rPower indique une polarité et donc permet de savoir si on consomme ou si on injecte.  
 
 valeurI      = emon1.Irms;               //extract Irms into Variable 
 rPower       = emon1.realPower;          //extract Real Power into variable 
      
 emon1.calcVI(30,30);          // Calculate all. No.of half wavelengths (crossings), time-out  
                               // d'origine emon1.calcVI(20,2000) 
 if( rPower < -seuilPoff ) { dim = dimmax; }   // arrêt en cas de chutte brusque de production  
 
 else {                        // test de polarité => puissance positive en mode injection 
   if( rPower > seuilP ) {     // l'injection augmente, on diminue le délai d'allumage du triac 
      if(dim > 0) {dim --; } else { dim = 0;}}        // que le délai ne devienne pas négatif ! 
      else if( rPower < -seuilP ) {                   // moins de prod : on baisse la charge 
         if(dim < dimmax) {dim ++; } else { dim = dimmax; }} 
 }   
 
  if (dim < 1) { digitalWrite(limiteLED, HIGH); }    // led témoin de surcharge 
  else { digitalWrite(limiteLED, LOW); } 
  analogWrite(triacLED, dimmax-dim);                 // Write the value to the LED for testing 
 
 // affichage de ce qui se passe / à décommenter dans le cadre du debugging  
   
    Serial.print("   "); 
    Serial.print(valeurI); 
    Serial.print("  |  "); 
    Serial.print(rPower); 
    Serial.print("  ||  "); 
    Serial.print(dim); 
    Serial.print("  |  "); 
    Serial.println(i); 
 
}                              // fin de Main Loop 

Illustration des essais 
 
Les essais ont été réalisés avec : 

 seuilP = 3000, ce qui implique une variabilité de 6000 de P (6Watts) 

 emon1.calcVI(30,30) soit acquisition sur 15 périodes, et mesure toutes les 30ms 
 
Remarque : il y a un compromis entre rapidité de réaction au changement de charge et/ou de production et stabilité 
des mesures. En choisissant un temps d’acquisition plus long, le temps de réaction est proportionnellement plus lent, 
mais les données récoltées moins sujettes à variations. 
 

Cas initial : Il n’y a rien côté production, charge de 160W 
 

La console indique : 

 
I entre 33 et 40  P entre -2000 et +2000  dim = 128 
i = 0 car avec dim à 128 il n’y a pas de décomptage de fire du triac 
Le wattmètre est branché sur la charge : là bien sûr il indique 0 

Cas 1 : on simule une production de 25W, charge de 160W 
 
I entre 240 et 400 P entre -4000 et +4000   dim = 97 
Wattmètre = 22W 
 
Remarque : si la charge « monte » à fond, il suffit de retourner la sonde de courant… 
 

 
 

Cas 2 : on simule une production de 85W, charge de 160W 
 
I entre 260 et 290 P entre -10000 et +9000  dim = 56 
Wattmètre = 85W 
 

 
 
  



21 

 

Cas 3 : on simule une production de 85W, charge de 60W 
 
I entre 140 et 160  P entre -27000 et +30000  dim = 0 
Wattmètre = 59W 
On remarque la LED d’overflow qui est allumée. 
On remarque également que la variation des mesures est bien plus faible que dans les 2 cas précédents, ici le triac 
étant déclenché dès le début, le pic de courant est quasi inexistant. 

 
 

Cas 4 : on simule une production de 25W, charge de 60W 
 
I entre 160 et 300  P entre -4000 et +4000   dim = 78 
Wattmètre = 26W 
 
Cas similaire au cas 1 sauf que la charge est réduite de moitié. La dim fait déclencher plus tôt pour compenser une 
résistance de charge plus élevée. 

Illustration de mise en boite 
 

 
 
 
 

 


